HP-15C Nth-degree Polynomial Fitting
Valentín Albillo (Ex-PPC #4747, HPCC #1075)

It frequently happens, both in theoretical problems and in real-life applications, that
we’ve got a series of data points (x,y) and we need to fit some smooth curve, defined
by a mathematical function, that passes through all of them. Once we’ve got a suitable
function, we can then perform such things as interpolation, inverse interpolation,
integration, differentiation, etc. Among the many functions we could fit to the dat a,
polynomials are by far the easiest to evaluate and manipulate, either numerically or
symbolically, so our problem is reduced to this: given a set of n+1 data points (x0 , y0 ),
…, (xn , yn ), find an nth-degree polynomial,
P(x) = a0 + a1 x + a2 x2 + a3 x3 + ... + an xn
such that P(xi ) = yi for i = 0, 1, …, n. This results in the following system of n+1
linear equations to determine the unknown coefficients a0 , a1 , ..., an :
a0 + a1 x0 + a2 x0 2 + a3 x0 3 + ... + an x0 n = y0
a0 + a1 x1 + a2 x1 2 + a3 x1 3 + ... + an x1 n = y1
... ... ... ... ... ... ...
a0 + a1 xn + a2 xn 2 + a 3 xn3 + ... + an xn n = yn
which we’ll construct and then solve using the powerful matrix capabilities of the
HP-15C. The maximum size for n is bounded by available memory, which for the 15C
amounts to a maximum of 64 registers available to store matrix elements.
For an nth-degree polynomial, the above system requires one (n+1)*(n+1) matrix and
a column vector which will hold both the yi data and then the ai coefficients. Thus, the
memory requirements to fit an nth degree polynomial are such that (n+1)*(n+2) <=
64.
The maximum value of n which satisfies the expression above is 6, so we can fit a 6thdegree polynomial, which requires 7*8 = 56 registers, thus leaving only 8 registers
for the program and any auxiliary registers needed. Assuming we’ll use at least one
extra storage register for loops, etc., this leaves just 7 registers = 49 bytes for the
program itself, and as some matricial instructions take more than 1 byte, this means
we have less than 49 steps for the program. Can it be done ? Yes !

The program
Here is a very small program for the HP-15C that I wrote specifically for this article.
It’s exactly 49 bytes long (including an implied RTN instruction at the very end). It’ll
prompt for data input, construct and solve the system, and output the resulting
coefficients, in as few as 42 program lines.

Program listing
• Very Important: steps 20, 23, 32, 38 and 41 must be entered in USER
mode, while all the rest must be entered out of USER mode. An “u”-like
character must appear next to the step number only for steps 20, 23, 32, 38
and 41, and no others.
01
02
03
04
05
06
07
08
09
10
11
12
13
14

LBL A
1
+
STO 2
ENTER
DIM A
1
DIM B
MATRIX 1
LBL 0
RCL DIM B
STO I
RCL 0

15 R/S
16 ENTER
17 ENTER
18 ENTER
19
1
20 uSTO A
21 LBL 1
22
*
23 uSTO A
24 LBL 2
25 DSE I
26 GTO 1
27 DSE 2
28 GTO 0

29 LBL B
30 RCL 0
31 R/S
32 uSTO B
33 GTO B
34 RCL MATRIX B
35 RCL MATRIX A
36 RESULT B
37
÷
38 uRCL B
39 LBL 3
40 R/S
41 uRCL B
42 GTO 3

Data Register usage
RI
R0
R1
R2

loop columns
row index for matrix elements
column index for matrix elements
loop rows

Notes
• steps 01-09 take the degree in X and dimension both matrices depending on it.

• steps 10-28 fill up the matrix row by row, setting up an internal loop (steps 21-26)
to compute and store all required powers of xi . Notice how the single user STO at
step 23 does most of the work.
• once all xi have been input, steps 29-33 are a very tight loop to prompt for the yi
and store them in the vector. The user STO at step 32 does triple duty by storing
each value in its proper place in the vector, updating the indexes, and testing for
loop termination.
• when all data points have been entered, steps 34-37 solve the system of equations,
with the simple-looking “division” operation at step 37 actually doing the
equivalent of a matrix inversion and subsequent matrix product with 13-digit
internal precision at microcode speeds (21 sec. for a 7x7 system). The matrix is
left in LU-decomposed form, so allowing for the fast computation of the
coefficients of another polynomial with the same xi but a different set of yi .
• finally, the loop at steps 39-42 (actually, a RCL twin of the STO one at steps 2933) outputs the computed coefficients, one at a time.

Usage instructions
1) After keying in the program, make sure you’re not in complex mode (press CF
8 if in doubt), then you must commit enough storage registers to the common
pool for the matrix operations. To that effect, press:
N, f DIM (i)
where N depends on the degree of the polynomial, as seen in this table:
Degree
1
2
3
4
5
6

N, f DIM (i)
2<= N <=52
2<= N <=46
2<= N <= 38
2<= N <=28
2<= N <=16
N = 2 (exact)

MEM
52 6 7-0
46 12 7-0
38 20 7-0
28 30 7-0
16 42 7-0
2 56 7-0

DIM A
2x2
3x3
4x4
5x5
6x6
7x7

DIM B
2x1
3x1
4x1
5x1
6x1
7x1

Note: if you get an error, this means you have registers commited to
other matrices: redimension the ones you don't need to 0x0 or even
easier, redimension all matrices to 0x0 executing MATRIX 0.
For instance, if you're going to fit a 3rd-degree polynomial to a set of 4 data
points, you may use 38, f DIM (i), but any number from 2 to 38 will do as well.
On the other hand, if you're going to fit a 6th-degree polynomial, you must use

2, f DIM (i). If you don't plan to use any other matrices or registers, you might
simply execute 2, f DIM (i) in all cases.
2) Now, enter the degree of the polynomial you want to fit (1 to 6), and start the
program by pressing either:
f PRGM, R/S or GSB A or (in User mode) A

à

1.0000

the program will prompt you to enter the x value of the 1st data point. (you
must enter all the x values first, then the y values):
x1 ,
x2 ,

R/S
R/S
...
xn+1 , R/S

à

2.0000
3.0000

à

1.0000

à

3) Now the program is prompting you to enter the y value of the 1st data point:
y1 ,
y2 ,
yn+1,

R/S à 2.0000
R/S à 3.0000
...
R/S à a0 , R/S à a1 , ... , R/S à an

and the polynomial which fits all n+1 data points is:
y = a 0 + a 1 x + a2 x2 + a3 x3 + ... + an xn
4) If you want to display all the coefficients again, set User mode and press:
f MATRIX 1
RCL B à a0 , RCL B à a1 , ... , RCL B à an
4) If you want to compute the coefficients for another set of data with the same xi
values but different yi values, you don’t need to reenter the x values. Press:
GSB B or (in User mode) B

à

1.0000

and go to (3) above to enter the new yi values. You'll notice that it takes much less
time to compute the coefficients now, as the matrix is already in LU-decomposed
form, and thus can be used as is, saving a lot of processing time.

Example 1: Data fitting and predictions
A series of measurements has resulted in the following table of values
corresponding to one independent variable x, and two dependent variables y, z. As
there are 7 triplets in all, fit a couple of 6th-degree polynomials y=y(x) and z=z(x)
to the data and use them to predict the values of y(x) and z(x) for x =-1.0, 0.0, +1.0
Var.
1st
2nd
3rd
4th
5th
6th
7th
X
-1.3118 -0.5663 -0.1503 +0.0323 +0.6236 +0.9312 +1.4202
Y
-0.9666 -0.5365 -0.1497 +0.0323 +0.5840 +0.8023 +0.9887
Z +0.2561 +0.8439 +0.9887 +0.9995 +0.8118 +0.5969 +0.1500
As you can see, the x values aren’t equally spaced, and although in this particular
example they're input in order of increasing value, that's not necessary either. So,
we’re going to compute the ai ,bi coefficients for these two 6th-degree polynomials:
y(x) = a0 + a1 x + a2 x2 + a3 x3 + ... + a6 x6
z(x) = b0 + b1 x + b 2 x2 + b3 x3 + ... + b 6 x6
First, we need to commit enough registers and clear all matrices by executing:
2, f DIM (i), f MATRIX 0
Now, input the degree (6), start the program, and at the prompts, enter all data points
for the 1st polynomial, first the x values, then the y values. In User mode, and FIX 5:
6, A
à 1.00000
-1.3118, R/S à 2.00000
-0.5663, R/S à 3.00000
... ... ... ... ... ... ... ...
1.4202, R/S à 1.00000
-0.9666, R/S à 2.00000
... ... ... ... ... ... ... ...
0.9887, R/S à 0.00001
R/S à 0.99985
... ... ... ... ... ... ... ...
R/S à 0.00027

[prompt to enter x 1 ]
[prompt to enter x 2 ]
[prompt to enter x 3 ]
[prompt to enter y 1 ]
[prompt to enter y 2 ]
[a0 coefficient (after just 25 seconds)]
[a1 coefficient]
[a6 coefficient]

so we've got our first polynomial, y(x):
y = 0.00001+0.99985 x+0.00021 x2-0.16592 x3 -0.00062 x4+0.00759 x5 +0.00027 x 6

Interpolated values are y(-1.0) = -0.84164, y(0.0) = 0.00001, and y(1.0) = 0.84139
To fit the z values, as the x values are the same, we don't need to reenter them, we just
need to input the z values. Assuming we’re still in User mode and FIX 5:
B
à 1.00000
[prompt to enter z1 ]
0.2561, R/S à 2.00000 [prompt to enter z 2 ]
... ... ... ... ... ... ... ... ... ...
0.1500, R/S à 1.00001 [b0 coefficient (after only 9 seconds !)]
R/S à 0.00026 [b1 coefficient]
... ... ... ... ... ... ... ... ... ...
R/S à -0.00174 [b6 coefficient]
and thus we've got our second polynomial, z(x) (only much faster !):
z = 1.00001+0.00026 x-0.50021 x2 -0.00089x3 +0.04248 x4+0.00044 x5 -0.00174 x6
Interpolated values are z(-1.0) = 0.54073, z(0.0) = 1.00001, and z(1.0) = 0.54035
The original data points were taken from y=sin(x) and z=cos(x), and now we can check
that our interpolated results are indeed correct to 4 decimal places as well.

Example 2: Constructing a polynomial knowing its roots
Find a 6th-degree polynomial P(x) such that its roots are 1/3, 1, sqrt(5), Pi, e,
sqrt(10) and P(Ln 2) = sin(1) (“sqrt” is the square root function)
In User mode, Rad mode, and FIX 5, the keystroke sequence:
6, A,
3, 1/X, R/S, 1, R/S, 5, Sqrt, R/S, Pi, R/S, 1, e x , R/S, 10, Sqrt, R/S,
2, Ln, R/S, 0, R/S, 0, R/S, 0, R/S, 0, R/S, 0, R/S, 0, R/S, 1, Sin, R/S
gives:
P(x) = -8.12096 + 44.25624 x -77.80622 x2 +62.24622 x3
-25.25201 x4 +5.08018 x5 -0.40346 x6

Example 3: Inverse interpolation
Find a 5th-degree polynomial x(k) which will approximate the root of x 3+ x - k = 0
for values of k between 2 and 10. Predict the value of the root x when k = 5.
First we evaluate k = x 3 +x for x = 1.0, 1.2, 1.4, 1.6, 1.8 and 2.0, to obtain these six
data points (k, x):
(2.0, 1.0), (2.928, 1.2), (4.144, 1.4), (5.696, 1.6), (7.632, 1.8) and (10.0, 2.0).
Notice we've reversed the order of x and k, because we're not interested in the value
of k given x, but in the value of x given k. Now, proceed as above to find the 5thdegree polynomial which fits those data points. In FIX 5, our program gives:
x(k) = 0.28469 + 0.50634 k -0.09666 k 2 +0.01292 k3 -0.00094 k4 + 0.00003 k5
which, for k=5 gives x = 1.5158. As the real positive root of x 3 + x – 5 = 0 comes out
as x = 1.5160, our polynomial actually gave nearly five significant digits correct.

Final remarks
It's amazing how the advanced capabilities and programming features of the HP -15C
allow it to fast and easily solve, in just a few program steps, problems that would take
hundreds of program lines and much longer execution times in all other
programmable calculators of its time. Even today, despite the tremendous advances in
technology, it still demonstrates what you can do with a machine so small it fits any
pocket size and which keeps on running long after the Duracell bunny has rusted away !
Note: For an HP-41C/Advantage implementation, see my article “Long Live the HP-41C
Advantage !” elsewhere in this issue. There you’ll find a thorough discussion on the ability
of the Advantage ROM to make available the advanced matrix capabilities of the HP15C in the HP-41C realm.

